**Abstract**

Hedeker and Nordgren (2013) present the stand-alone MIXREGLS program for fitting the mixed-effects location scale model to continuous longitudinal and other clustered data. This model can be used when interest lies in joint modeling the mean and dispersion of subjects’ responses over time. The model extends the standard two-level random-intercept mixed model by allowing both the within- and between-subject variances to be influenced by the covariates and for the within-subject variance to additionally depend on a subject random-scale effect. In this article we present the runmixregrs command to run MIXREGLS seamlessly from within Stata. We illustrate the notable advantages of using runmixregrs by replicating and extending the two example analyses presented in Hedeker and Nordgren (2013). We then use runmixregrs to demonstrate a new and important research finding. Namely, that ignoring the random-scale effect in the within-subject variance function will lead to the regression coefficients in this function to be estimated with spurious precision, especially the regression coefficients of subject-level covariates.

**Keywords:** mixed-effects location scale model, mixed-effects model, mixed model, multilevel model, hierarchical linear model, random effects model, heterogeneity of variance, heteroskedasticity, variance modeling, variance functions, MIXREGLS, Stata, runmixregrs.

1. **Introduction**

Mixed-effects models (Fitzmaurice, Laird, and Ware 2011; Hedeker and Gibbons 2006; Verbeke and Molenberghs 2000) – also known as multilevel models (Goldstein 2011; Snijders and Bosker 2012), hierarchical linear models (Raudenbush and Bryk 2002), or random effects models – are widely used in the analysis of longitudinal and other clustered data in
the social, behavioral and medical sciences. The simplest mixed-effects model for continuous repeated measures data, where observations are nested within subjects, is the two-level random-intercept model. This model decomposes the variation in the response variable, having adjusted for the covariates, into a homogeneous between-subject (BS) variance and a homogeneous within-subject (WS) variance.

Hedeker and Nordgren (2013) propose the mixed-effects location scale model to extend the above model by modeling both the WS and BS variances as log-linear functions of the covariates. A subject random-scale effect is then entered into the WS variance function to account for any remaining unexplained heterogeneity in the residual error variance between subjects. The model is argued to be especially useful when interest lies in joint modeling the mean and dispersion of subjects’ responses over time. However, it can equally be employed to analyze clustered data when interest lies in explaining why both the mean and dispersion of subjects’ outcomes vary across clusters (e.g., hospitals, schools, neighborhoods, etc.).

Hedeker, Mermelstein, and Demirtas (2008b) show how mixed-effects location scale models can be fitted using maximum likelihood estimation in SAS (SAS Institute Inc 2012) via SAS PROC NL MIXED. However, Hedeker and Nordgren (2013) note that this procedure can be slow to run, often requires excellent starting values to converge, and is difficult for less statistically literate data analysts to use. It is these difficulties which motivated the development of their stand-alone MIXREGLS program (Hedeker and Nordgren 2013). MIXREGLS is Windows-based and is run in batch mode; there is no graphical user interface. A limitation of MIXREGLS is that it has no facilities beyond this single estimation capability. Furthermore, the process of writing MIXREGLS data and model definition files is fiddly and error prone. Hedeker and Nordgren (2013) recognize these difficulties and show how the process of fitting models in MIXREGLS can be partially automated using SAS and to a fuller extent using R. However, no parallel functionality is provided for the Stata software (StataCorp. 2013).

Other software can also fit mixed-effects location scale models. Rast, Hofer, and Sparks (2012) present syntax to fit these models using Markov chain Monte Carlo (MCMC) methods in WinBUGS (Lunn, Thomas, Best, and Spiegelhalter 2000). We note that WinBUGS can itself be called from Stata using the vb (Thompson, Palmer, and Moreno 2006) and vbs (Thompson 2014) suites of commands. However, WinBUGS is also difficult for less statistical literate data analysts to use. Leckie, French, Charlton, and Browne (2015) discuss fitting similar models using MCMC methods in the Stat-JR statistics package (Charlton, Michaelides, Parker, Cameron, Szmargad, Yang, Zhang, Frazer, Goldstein, Jones, Leckie, Moreau, and Browne 2013), however Stat-JR is not currently accessible within Stata. Certain mixed-effects location scale models can be formulated using the double hierarchical generalized linear model (DHGLM) framework proposed by Lee and Nelder (2006) and are implemented in ASReml (Gilmour, Gogel, Cullis, and Thompson 2009), GenStat (Payne, Murray, Harding, Baird, and Soutar 2009) and R (R Core Team 2014) (via the dhglm package; Noh and Lee 2013), but none of these packages are accessible from within Stata. It is worth noting that it is also not currently possible to fit these models using inbuilt Stata commands. Nor is it possible to fit these models in SPSS (IBM Corporation 2012) or in specialized mixed-effects modeling software such as HLM (Raudenbush, Bryk, and Congdon 2012), MLwiN (Rasbash, Charlton, Browne, Healy, and Cameron 2009) and SuperMix (Hedeker, Gibbons, du Toit, and Cheng 2008a).

In this article, we present the runmixregls command to run the MIXREGLS software seamlessly from within Stata. There are notable advantages to running MIXREGLS in this way.
First, runmixregls provides a simple and intuitive command syntax which allows users to specify the full range of modeling and estimation options implemented in MIXREGLS. Second, prior to fitting a particular model, users can take advantage of Stata’s excellent statistics, graphics and data management commands to prepare and descriptively analyze their data. Third, after fitting the model, users can apply Stata’s comprehensive hypothesis testing, model comparison, prediction and plotting facilities to interpret their models and to communicate their results. Last, all analyses can be reproduced and documented for publication and review by typing all these commands into a file and running them all at once.

The remainder of the article is structured as follows. Section 2 briefly reviews the mixed-effects location scale model implemented in MIXREGLS. Section 3 describes how to install runmixregls from within Stata. Section 4 presents the runmixregls command syntax and describes the various modeling and estimation options. Sections 5 and 6 illustrate runmixregls by replicating and extending the two example analyses presented in Hedeker and Nordgren (2013). Section 7 performs a simulation study using runmixregls to demonstrate a new and important research finding. Namely, that ignoring random-scale effects in the WS variance function will lead the WS variance function regression coefficients to be estimated with spurious precision, especially the regression coefficients of subject-level covariates. Section 8 concludes.

2. Review of the mixed-effects location scale model

2.1. Mixed-effects model

Let \( y_{ij} \) denote the continuous response measurement for subject \( i \) \( (i = 1, 2, ..., N) \) at occasion \( j \) \( (j = 1, 2, ..., n_i) \). The standard two-level random-intercept mixed-effects model can then be written as

\[
    y_{ij} = x_{ij}^\top \beta + v_i + \epsilon_{ij},
\]

\[
    v_i \sim \mathcal{N}(0, \sigma^2_v),
\]

\[
    \epsilon_{ij} \sim \mathcal{N}(0, \sigma^2_\epsilon),
\]

where \( x_{ij} \) is a vector of covariates, \( \beta \) is the associated vector of coefficients, \( v_i \) is the random-intercept effect, and \( \epsilon_{ij} \) is the residual error. The covariates may be time varying or time invariant. The random-intercept effect and residual error are assumed normally distributed with zero means and constant variances. The homogeneous BS variance, \( \sigma^2_v \), measures the variability in subjects’ mean responses, having adjusted for the covariates. The homogeneous WS variance, \( \sigma^2_\epsilon \), measures the variability in subjects’ measurements about their adjusted mean responses. This model can be fitted in Stata using the \texttt{xtreg} command with the \texttt{mle} option, or alternatively by using the \texttt{mixed} command.\footnote{Prior to Stata 13 (released June 2013), the \texttt{mixed} command was named \texttt{xtmixed}.}

2.2. Mixed-effects location scale model

The mixed-effects location scale model implemented in MIXREGLS may be viewed as an
extended reparameterized version of the above model and can be written as

\[ y_{ij} = x_{ij}^\top \beta + \sigma_{v_{ij}} \theta_{1i} + \epsilon_{ij}, \quad (4) \]

\[ \log \left( \sigma_{v_{ij}}^2 \right) = u_{ij}^\top \alpha, \quad (5) \]

\[ \log \left( \sigma_{e_{ij}}^2 \right) = w_{ij}^\top \tau + \tau_1 \theta_{1i} + \tau_q \theta_{2i}^2 + \sigma_\omega \theta_{2i}, \quad (6) \]

\[ \theta_{1i} \sim \mathcal{N}(0, 1), \quad (7) \]

\[ \theta_{2i} \sim \mathcal{N}(0, 1), \quad (8) \]

\[ \epsilon_{ij} \sim \mathcal{N}\left(0, \sigma_{e_{ij}}^2\right), \quad (9) \]

where we refer to Equation 4, Equation 5 and Equation 6 as the mean function, the BS variance function, and the WS variance function, respectively.

Hedeker and Nordgren (2013, p. 3) describe this model in detail and so we provide only a brief summary here.

The mean function (Equation 4) is the same as that in the standard model (Equation 1), except that the random-intercept effect, now referred to as the random-location effect, is parameterized in standardized form, \( \sigma_{v_{ij}} \theta_{1i} \). The first term, \( \sigma_{v_{ij}} \), denotes the square root of the BS variance, while \( \theta_{1i} \) denotes the standardized random-location effect, \( \theta_{1i} \sim \mathcal{N}(0, 1) \).

Note that \( \sigma_{v_{ij}} \) is subscripted by \( i \) and \( j \) to indicate that its value may change across subjects and occasions. The influence of \( \theta_{1i} \) on \( y_{ij} \) may therefore be amplified or dampened by the magnitude of \( \sigma_{v_{ij}} \).

The BS variance function (Equation 5), models the BS variance, \( \sigma_{v_{ij}}^2 \), as a log-linear function of a second vector of subject- or occasion-level covariates, \( u_{ij} \), where \( \alpha \) denotes the associated vector of coefficients.

The WS variance function (Equation 6), models the WS variance, \( \sigma_{e_{ij}}^2 \), as a log-linear function of a third vector of subject- or occasion-level covariates, \( w_{ij} \), where \( \tau \) is the associated vector of coefficients. A quadratic subject-level association is allowed between the unexplained location and scale variability by entering \( \theta_{1i} \) and its square, \( \theta_{1i}^2 \), into the WS variance function as latent covariates with regression coefficients, \( \tau_1 \) and \( \tau_q \), to be estimated. This additional flexibility is useful when the response exhibits floor or ceiling effects, as we then expect a concave relationship between subjects’ variances and means whereby subjects with very low or very high means have near-zero WS variances, while subjects with means closer to the middle of the response scale have higher WS variances. A quadratic association is better able to capture such concavity. Finally, a new random effect, denoted \( \theta_{2i} \), and referred to as a random-scale effect, is included to account for unexplained variation in the WS variance above and beyond the contribution of the covariates. This random effect is assumed normally distributed with zero mean and constant variance, \( \sigma_{\omega}^2 \).

When \( u_{ij} \) and \( w_{ij} \) each include only a constant and when \( \tau_1 = \tau_q = \sigma_\omega = 0 \), the above mixed-effects location scale model simplifies to a reparameterized version of the standard two-level random-intercept mixed-effects model with homogeneous variances presented in Section 2.1.

The use of log-link functions ensures positive variances. However, it makes parameter interpretation less straightforward. In particular, the covariates have multiplicative rather than additive effects on the variances. In the examples we plot the predicted level-1 variance functions to aid their substantive interpretation. This proves especially helpful in interpreting quadratic random-location effects on the within-subject variance.
As in standard mixed-effects models, the random effects normality assumptions may not necessarily hold and it will be prudent to check their plausibility, for example, by inspecting quantile-quantile (Q-Q; normal scores plots) or other graphical plots post-estimation.

3. Installing runmixregls

runmixregls requires Stata 12 or later and MIXREGLS. MIXREGLS can be freely downloaded from: http://www.uic.edu/sph/downloads/cms-downloads/cbd/zipfiles/MIXREGLS.zip.

runmixregls can be installed from the Statistical Software Components (SSC) archive by typing the following command within a net-aware version of Stata

```
.ssc install runmixregls
```

Two files will be installed on your computer: runmixregls.ado, a Stata ado file which defines the command; and runmixregls.sthlp, a Stata help file which documents the command. Note that these files will be installed onto your adopath, the path where Stata searches for the files it needs.

If you have already installed runmixregls from the SSC, you can check that you are using the latest version by typing the following command

```
.adoupdate runmixregls
```

Next, you must declare the fully qualified filename for the MIXREGLS executable (the MIXREGLS.exe file) so that runmixregls knows where to find the software. You can do this by specifying a global macro called mixreglspath. Users must substitute the fully qualified filename that is correct for them, e.g.,

```
.global mixreglspath "C:\MIXREGLS\mixreglsb.exe"
```

Advanced users may wish to set the MIXREGLS fully qualified filename every time Stata is started by simply inserting the following line into their profile do-file profile.do (see help profile).

4. Syntax diagram

The runmixregls command follows standard Stata syntax for estimation commands. We restrict our discussion here to the most common options. A complete description is provided in the runmixregls help file.\(^2\) You may type the following at any point to view this help file

```
.help runmixregls
```

\(^2\)The runmixregls help file is also available as a PDF document (Leckie 2013). Note that the terminology used in the help file follows Stata conventions and therefore differs slightly from that presented here where we have aimed to be consistent with Hedeker and Nordgren (2013).
Before using `runmixregls`, you must declare the subject identifier (the variable in the data which uniquely identifies the subjects) using the `xtset` command (see `help xtset`).

```
xtset varname
```

where `varname` is the name of the subject identifier.

### 4.1. Syntax

The `runmixregls` command has the following syntax

```
runmixregls depvar [varlist] [if] [in] [, options]
```

where `runmixregls` is the name of the command, `depvar` denotes the dependent or response variable, `varlist` denotes the list of covariates appearing in the mean function (the square brackets indicate optional arguments), the `if exp` qualifier allows one to restrict the estimation sample to those observations for which the expression `exp` is true; the `in range` qualifier allows one to restrict the estimation sample to the range of observations given by `range`, the comma separates the specification of the mean function from the specification of any modeling or estimation options listed in `options`.

### 4.2. Options

The `options` are defined as follows

**Model**

- `noconstant` suppresses the constant term (intercept) in the mean function.
- `between(varlist[, noconstant])` specifies the variables in the BS variance function.
- `within(varlist[, noconstant])` specifies the variables in the WS variance function.
- `association(etype)`, where `etype` can be `none|linear|quadratic`
  - specifies the subject-level association between the (log of the) WS variance and the random-location effect. The default is `association(linear)`.

**Random effects/Residuals**

- `reffsets(newvar1 newvar2)` retrieves the empirical Bayes estimates of the standardized random effects. The standardized random-location effects are placed in `newvar1` while the standardized random-scale effects are placed in `newvar2`. The associated standard errors are placed in `newvar1_se` and `newvar2_se`.
- `residuals(newvar)` retrieves the standardized residual errors and places them in `newvar`.

**Integration**

- `noadapt` prevents MIXREGLS from using adaptive Gaussian quadrature. MIXREGLS will use ordinary Gaussian quadrature instead.
intpoints(#) sets the number of integration points for (adaptive) Gaussian quadrature. The default is intpoints(11). The more points, the more accurate the approximation to the log-likelihood. However, computation time increases with the number of quadrature points. When models do not converge properly, increasing the number of quadrature points can sometimes lead to convergence.

Maximization

iterate(#) specifies the maximum number of iterations. The default is iterate(200). You should seldom have to use this option.

tolerance(#) specifies the convergence tolerance. The default is tolerance(0.0005). You should seldom have to use this option.

standardize standardizes all covariates in all functions during optimization. This ensures all covariates are on the same numerical scale with mean 0 and variance 1. This can be helpful if the model “blows up” or does not converge to the solution.

MIXREGLS model files

typedatfile displays the MIXREGLS model data file in the results window.
typedeffile displays the MIXREGLS model definition file in the results window.
typeoutfile displays the MIXREGLS model output file in the results window.
savedatfile(filename[, replace]) saves the MIXREGLS model data file. A fully qualified filename should be specified. The replace option overwrites the MIXREGLS model definition file if it already exists.
savedeffile(filename[, replace]) saves the MIXREGLS model definition file. A fully qualified filename should be specified. The replace option overwrites the MIXREGLS model definition file if it already exists.
saveoutfile(filename[, replace]) saves the MIXREGLS model output file. A fully qualified filename should be specified. The replace option overwrites the MIXREGLS output file if it already exists.

Reporting

level(#) specifies the confidence level, as a percentage, for confidence intervals. The default is level(95).

display_options controls column formats, row spacing, line width, and display of omitted variables and base and empty cells; see help estimation options##display_options for more details.

noheader suppresses the display of the summary statistics at the top of the output; only the coefficient table is displayed.

notable suppresses display of the coefficient table.

coefflegend specifies the legend of the coefficients and how to specify them in an expression to be displayed rather than displaying the statistics for the coefficients.
5. Example 1: Reisby depression study

The first example analyses presented by Hedeker and Nordgren (2013) relate to data drawn from a six-week longitudinal psychiatric study of 66 depressed inpatients (Reisby, Gram, Bech, Nagy, Petersen, Ortmann, Ibsen, Dencker, Jacobsen, Krautwald, Sondergaard, and Christiansen 1977). Patients were diagnosed at baseline with either endogenous ($N = 37$) or non-endogenous ($N = 29$) depression and were then rated weekly using the Hamilton depression rating scale. The data are two-level with observations (level-1) nested within patients (level-2). The total number of observation is 375, 21 less than the expected $66 \times 6 = 396$ as some subjects were missing on some weeks.

5.1. Data

We start by loading the data. We do this using the `use` command where we specify the `clear` option to replace any data should they currently exist in memory.

```
use reisby, clear
```

We then use the `codebook` command with the `compact` option to compactly describe the data contents.

```
codebook, compact
```

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Unique</th>
<th>Mean</th>
<th>Min</th>
<th>Max</th>
<th>Label</th>
</tr>
</thead>
<tbody>
<tr>
<td>id</td>
<td>396</td>
<td>66</td>
<td>323.9697</td>
<td>101</td>
<td>610</td>
<td>Subject</td>
</tr>
<tr>
<td>hamdep</td>
<td>396</td>
<td>37</td>
<td>16.22475</td>
<td>-9</td>
<td>39</td>
<td>Depression score</td>
</tr>
<tr>
<td>week</td>
<td>396</td>
<td>6</td>
<td>2.5</td>
<td>0</td>
<td>5</td>
<td>Week</td>
</tr>
<tr>
<td>endog</td>
<td>396</td>
<td>2</td>
<td>.5606061</td>
<td>0</td>
<td>1</td>
<td>Endogenous</td>
</tr>
<tr>
<td>endweek</td>
<td>396</td>
<td>6</td>
<td>1.40404</td>
<td>0</td>
<td>5</td>
<td>Endogenous * Week</td>
</tr>
</tbody>
</table>

The output shows that the data includes five variables: the subject identifier, `id`; the response variable, `hamdep` (range = 0 to 39, where -9 is the missing value code); the week number, `week` (coded sequentially from 0 to 5); the grouping variable, `endog` (coded 0 for non-endogenous subjects, 1 for endogenous subjects); and the interaction between `endog` and `week`, `endweek`. We use the `recode` command to recode the 29 missing values described above from -9 to Stata’s system missing value . so that these observations will not be used in the analysis.

```
recode hamdep (-9 = .)
```

(hamdep: 21 changes made)

To get a better sense of the data, we plot depression scores against time separately for non-endogenous and endogenous subjects. We use the `twoway` command with the `line` plottype to give a twoway line plot of `hamdep` against `week`. The `connect(ascending)` option requests that we connect the points in ascending order of the x variable, here `week`. This ensures that we plot a separate line for each subject. The `xlabel(0(1)5)` option specifies that major ticks and labels appear on the x axis from 0 to 5 in steps of 1. The `by(endog)` option requests we draw separate plots within the graph for the two subject groups defined by `endog`. 
Figure 1: Spaghetti plot of depression trajectories plotted separately for non-endogenous (left panel) and endogenous subjects (right panel).

```
    . twoway (line hamdep week, connect(ascending)), xlabel(0(1)5) by(endog)
```

The plot shows depression scores tend to improve over time for all subjects. Endogenous subjects appear to have higher depression scores at baseline but it is unclear as to whether these subjects tend to recover more or less rapidly than non-endogenous subjects. There is substantial variability between subjects and this increases over time; the observed trajectories fan outwards as the study progresses. There is a slight suggestion that endogenous subjects are more heterogeneous than non-endogenous subjects at each point in time, but this is far from certain. Some subjects certainly appear to recover more erratically than others, but it is less clear as to whether the erraticness of individual subjects’ sequences of depression scores is related to time or depression group.

5.2. Model

Hedeker and Nordgren (2013, p. 11) address these questions by specifying the following mixed-effects location scale model

\[
\text{hamdep}_{ij} = \beta_0 + \beta_1 \text{week}_{ij} + \beta_2 \text{endog}_i + \beta_3 \text{endweek}_{ij} + \sigma_v \theta_1 + \epsilon_{ij},
\]  

(10)

\[
\log \left( \sigma_{v_i}^2 \right) = \alpha_0 + \alpha_1 \text{endog}_i,
\]  

(11)

\[
\log \left( \sigma_{\epsilon_{ij}}^2 \right) = \tau_0 + \tau_1 \text{week}_{ij} + \tau_2 \text{endog}_i + \tau_3 \theta_1 + \sigma_\omega \theta_2,
\]  

(12)

\[
\theta_{1i} \sim \mathcal{N}(0, 1),
\]  

(13)

\[
\theta_{2i} \sim \mathcal{N}(0, 1),
\]  

(14)

\[
\epsilon_{ij} \sim \mathcal{N}(0, \sigma_{\epsilon_{ij}}^2),
\]  

(15)
where, in the mean function, \( \text{hamdep} \) is modeled in terms of \( \text{week} \), \( \text{endog} \) and \( \text{endweek} \), to allow endogenous and non-endogenous subjects to differ in baseline depression severity and to recover at different rates. The random-location effect, \( \sigma_{\theta_1i} \), allows for subject intercept heterogeneity above and beyond that explained by the covariates. The log of the BS variance, \( \sigma^2_{u_i} \), is modeled as a function of \( \text{endog} \) to allow endogenous and non-endogenous subjects to be differentially variable in terms of their mean depression scores, having adjusted for time, group and group by time interaction effects. The log of the WS variance, \( \sigma^2_{\epsilon ij} \), is modeled as a function of \( \text{week} \) and \( \text{endog} \) to allow the variability of a subject’s depression scores about their individual trajectories to differ across the two groups and to change over time. The location effect is entered into the WS variance function to allow for a linear subject-level association between the log of the WS variance and the random-location effect. The random-scale effect, \( \sigma_{\omega \theta_2i} \), allows for any remaining unexplained variation in WS response heterogeneity across subjects.

An important feature of the data shown in Figure 1, but not captured by the model is that subjects vary in terms of their time trends as well as their intercepts. Ideally we would introduce a random-slope on \( \text{week} \) into the model to account for this. However, this modeling extension is not currently implemented in MIXREGLS. Restricting subjects to have parallel trajectories will force the WS variance to increase as a function of time.

Before we can fit this model, we must first use the \texttt{xtset} command to declare the grouping or panel variable, which for these data is \( \text{id} \)

\[
\text{. \texttt{xtset id}}
\]

\[
\text{panel variable: id (balanced)}
\]

The above model can then be fitted using the following \texttt{runmixregls} command.

\[
\text{. \texttt{runmixregls hamdep week endog endweek, between(endog) within(week endog)}}
\]

\texttt{runmixregls} - Run MIXREGLS from within Stata

Mixed-effects location scale model

\[
\begin{array}{l}
\text{Number of obs} = 375 \\
\text{Number of groups} = 66 \\
\text{Obs per group: min} = 4 \\
\text{avg} = 5.7 \\
\text{max} = 6 \\
\end{array}
\]

Run time (seconds) = 5.715
Integration points = 11
Log Likelihood = -1122.2965

| hamdep | Coef. | Std. Err. | z   | P>|z|   | [95% Conf. Interval] |
|--------|-------|-----------|-----|-------|---------------------|
| Mean   |       |           |     |       |                     |
| week   | -2.295431 | .1877299 | -12.23 | 0.000 | -2.663375 to -1.927488 |
| endog  | 1.879423 | 1.076568  | 1.75 | 0.081 | -.2306119 to 3.989457 |
| endweek| -.028614 | .2677226  | -0.11 | 0.915 | -.5533407 to .4961127 |
The first term after the name of the command is the response variable, `hamdep`, and this is then followed by the list of mean function covariates, `week endog endweek`. The first option, `between(endog)`, specifies the BS variance covariates, the second option, `within(week endog)`, specifies the WS variance covariates.

The formatting of the `runmixregls` output follows that of other Stata panel data estimation commands. A range of ancillary model information is presented first, followed by the table of parameter estimates and standard errors.

The first line of output simply states that `MIXREGLS` was run from within Stata using the `runmixregls` command. Immediately below that, the output states that the model is a mixed-effects location scale model where the grouping or panel variable is `id`. To the right, the output states that the model is fitted to an estimation sample of 375 observations belonging to 66 subjects where the number of observations per subject ranges from 4 to 6 with an average of 5.7 observations.

The next block of output, immediately above the table, reports that `MIXREGLS` took 5.715 seconds to fit the model using 11 integration points. The log-likelihood statistic is also presented.

The table of parameter estimates and standard errors is presented in five blocks, labeled `Mean` (mean function regression coefficients), `Between` (BS variance function regression coefficients), `Within` (WS variance function regression coefficients), `Association` (subject-level location-scale association parameters) and `Scale` (random-scale effect standard deviation). The table also reports z-ratios, p values and 95% confidence intervals.

The substantive interpretation of these results is given by Hedeker and Nordgren (2013, p. 13), and so we provide only a brief summary here. The mean model shows a significant negative effect of time ($\hat{\beta}_1 = -2.29, p < 0.05$) indicating depression scores decrease over the six-week study in the non-endogenous group. The endogenous group does not differ significantly from this trend in either their mean baseline depression score ($\hat{\beta}_2 = 1.88, p = 0.08$) or in their average rate of improvement over time ($\hat{\beta}_3 = -0.03, p = 0.92$). There are also no group differences in terms of the BS variance ($\hat{\alpha}_1 = 0.51, p = 0.27$) or the WS variance ($\hat{\tau}_2 = 0.29$,
Thus, endogenous subjects are found to be no more or less heterogeneous than non-endogenous subjects in either their mean depression scores or in the variability of their depression scores about their individual trajectories means. The WS variance significantly increases with time ($\hat{\tau}_1 = 0.19, p < 0.05$) suggesting that a subject’s depression scores become more variable with time. However, as discussed above, the magnitude of this effect would be expected to reduce were we able to introduce random time trends into the model. The linear random-location effect does not have a significant effect on the log of the WS variance ($\hat{\tau}_l = 0.21, p = 0.14$). Finally, the scale standard deviation is significant ($\hat{\sigma}_w = 0.66, p < 0.05$) indicating that some subjects’ depression scores are significantly more dispersed than other subjects even after adjusting for group and time effects.

### 5.3. Random effects and residuals

Next we retrieve the empirical Bayes estimates of the standardized random-location and random-scale effects by specifying the `reffects(theta1 theta2)` option. The standardized random-location effects are placed in a new variable named `theta1` while the standardized random-scale effects are placed in `theta2`. The associated standard errors are placed in `theta1_se` and `theta2_se`. We retrieve the standardized residual errors by specifying `rstandard(estd)`.

```stata
. runmixregls hamdep week endog endweek, between(endog) within(week endog)
> reffects(theta1 theta2) residuals(estd)

(output omitted)
```

The `runmixregls` output is identical to before and so we do not redisplay it here.

Figure 2 in Hedeker and Nordgren (2013, p. 20) presents a histogram of the standardized model residuals which shows them to be approximately normally distributed. Our Figure 2 replicates this figure using the `histogram` command.

```stata
. histogram estd, width(0.5) start(-3) frequency
(bin=11, start=-3, width=.5)
```

where we specify the `width(0.5)` and `start(-3)` options to set the width of the bins to 0.5 and to set the theoretical minimum value to -3. The `frequency` option requests the histogram to be drawn as frequencies.

Figure 3 in Hedeker and Nordgren (2013, p. 21) presents a bivariate scatter plot of the estimated random-location and random-scale effects. Our Figure 3 replicates this figure using the `scatter` command.

```stata
. scatter theta2 theta1, mlabel(id)
```

where we have added the `mlabel(id)` option to add marker labels to the figure showing subjects’ unique identifier values, `id`. The plot shows no obvious association between the estimated location and scale effects corroborating the non-significance of the random-location effects on the WS variance ($\hat{\tau}_l = 0.21, p = 0.14$).
Tables 1 and 2 of Hedeker and Nordgren (2013, p. 18–20) present subjects who appear interesting in terms of their location and scale estimates. We can easily replicate these tables, but we must first carry out a small number of data manipulation steps.
We start by only keeping those variables needed to construct the tables and by renaming the response from `hamdep` to `hd`.

```
. keep id hamdep week theta1 theta2
. rename hamdep hd
```

Next we use the `reshape` command to convert the data from “long form” (one record per observation) to “wide form” (one record per subject).

```
. reshape wide hd, i(id) j(week)
```

We then reformat the display precision of `theta1` and `theta2` to three decimal places.

```
. format %4.3f theta1 theta2
```

Lastly we use the `gsort` command to sort the data in descending order by `theta2`.

```
. gsort -theta2
```

Table 1 in Hedeker and Nordgren (2013, p. 18) lists the two subjects with the highest scale estimates and the two subjects with the lowest scale estimates. We can replicate the first half of this table by using the `list` command where we specify `if inlist(_n, 1, 2)` to list the data for only the first two records (subjects) in the data (i.e., the two subjects with the highest scale estimates). These two subjects (Subjects 606 and 505) show unusual trajectories whereby their baseline depression scores appear somewhat anomalous relative to their subsequent scores.

```
. list id theta2 hd? if inlist(_n, 1, 2)
```

```
+--------------------------------------------------+
<table>
<thead>
<tr>
<th>id theta2 hd0 hd1 hd2 hd3 hd4 hd5</th>
</tr>
</thead>
</table>
1. | 606 1.585 19 33 12 12 3 1 |
2. | 505 1.532 21 11 18 0 0 4 |
+--------------------------------------------------+
```
We can replicate the second half of this table by changing the `if` statement to only list the last two records (subjects) in the data (i.e., the two subjects with the lowest scale estimates). These two subjects (Subjects 335 and 308) have linear rates of decline which closely match those predicted by the mean time trend.

```
. list id theta2 hd? if inlist(_n, 65, 66)
```

```
+--------------------------------------------------+
<table>
<thead>
<tr>
<th>id    theta2   hd0   hd1   hd2   hd3   hd4   hd5</th>
</tr>
</thead>
<tbody>
<tr>
<td>65.</td>
</tr>
<tr>
<td>66.</td>
</tr>
</tbody>
</table>
+--------------------------------------------------+
```

Table 2 in Hedeker and Nordgren (2013, p. 20) presents the data for subjects located in the four corners of Figure 2. To replicate this table, we first sort the data in ascending order by `theta2`.

```
. sort theta2
```

Subjects 117 and 347 are located in the bottom left corner of Figure 2. These subjects have some of the mildest depression scores and show continual smooth improvement in their scores over the course of the study.

```
. list id theta1 theta2 hd? if inlist(id, 117, 347)
```

```
+-----------------------------------------------------------+
<table>
<thead>
<tr>
<th>id    theta1    theta2   hd0   hd1   hd2   hd3   hd4   hd5</th>
</tr>
</thead>
<tbody>
<tr>
<td>11.</td>
</tr>
<tr>
<td>43.</td>
</tr>
</tbody>
</table>
+-----------------------------------------------------------+
```

Subject 345 is located in the bottom right corner of Figure 2. This subject also shows continual smooth improvement in their scores over the course of the study, but these scores are always substantially above average.

```
. list id theta1 theta2 hd? if inlist(id, 345)
```

```
+-----------------------------------------------------------+
<table>
<thead>
<tr>
<th>id    theta1    theta2   hd0   hd1   hd2   hd3   hd4   hd5</th>
</tr>
</thead>
<tbody>
<tr>
<td>41.</td>
</tr>
</tbody>
</table>
+-----------------------------------------------------------+
```

Subject 505 is located in the top left corner of Figure 2. This subject shows a very erratic series of mostly low depression scores.
. list id theta1 theta2 hd? if inlist(id, 505)

<table>
<thead>
<tr>
<th>id</th>
<th>theta1</th>
<th>theta2</th>
<th>hd0</th>
<th>hd1</th>
<th>hd2</th>
<th>hd3</th>
<th>hd4</th>
<th>hd5</th>
</tr>
</thead>
<tbody>
<tr>
<td>58</td>
<td>505</td>
<td>-1.320</td>
<td>1.532</td>
<td>21</td>
<td>11</td>
<td>18</td>
<td>0</td>
<td>0</td>
</tr>
</tbody>
</table>

Subjects 607, 322, 328 and 360 are located in the top right corner of Figure 2. These subjects show erratic patterns of depression scores which are mostly substantially above average and show little improvement with time.

. list id theta1 theta2 hd? if inlist(id, 607, 322, 328, 360)

<table>
<thead>
<tr>
<th>id</th>
<th>theta1</th>
<th>theta2</th>
<th>hd0</th>
<th>hd1</th>
<th>hd2</th>
<th>hd3</th>
<th>hd4</th>
<th>hd5</th>
</tr>
</thead>
<tbody>
<tr>
<td>57</td>
<td>607</td>
<td>1.517</td>
<td>0.919</td>
<td>30</td>
<td>39</td>
<td>30</td>
<td>27</td>
<td>20</td>
</tr>
<tr>
<td>58</td>
<td>322</td>
<td>1.272</td>
<td>0.946</td>
<td>28</td>
<td>21</td>
<td>25</td>
<td>32</td>
<td>34</td>
</tr>
<tr>
<td>60</td>
<td>328</td>
<td>1.676</td>
<td>0.992</td>
<td>22</td>
<td>24</td>
<td>28</td>
<td>26</td>
<td>28</td>
</tr>
<tr>
<td>61</td>
<td>360</td>
<td>1.333</td>
<td>1.003</td>
<td>21</td>
<td>28</td>
<td>27</td>
<td>29</td>
<td>28</td>
</tr>
</tbody>
</table>

5.4. Additional analyses extending those of Hedeker and Nordgren (2013)

In Section 7 we shall use `runmixregls` in a simulation study which presents a new and important research finding. Namely, that ignoring random-scale effects in the WS variance function will lead the WS variance function regression coefficients to be estimated with spurious precision, especially the regression coefficients of subject-level covariates. In the remainder of this section we shall motivate this simulation study by examining the impact of ignoring random-scale effects in the Reisby data.

We must first reload the data, recode the missing values of `hamdep`, and declare the data to be panel data.

. use reisby, clear
. recode hamdep (-9 = .)

(hamdep: 21 changes made)

. xtset id

    panel variable: id (balanced)

Next we refit the model, removing the non-significant association between the WS variance and the random-location effects. We increase the number of integration points to 15 simply to demonstrate how one would do this.
The only subject-level covariate in the WS variance function is the depression group variable. The results show that there is no significant difference in WS variance between the endogenous and non-endogenous subjects, even at the 20% level ($\hat{\tau}_2 = 0.30, p = 0.22$). We shall compare this finding with that obtained from a model which ignores the significant random-scale effects. At this point it is helpful to understand that MIXREGLS has fitted the full model shown above in three sequential and increasingly complex stages. The results of each model stage are provided as the starting values for the next stage. This sequential approach improves the convergence of the full model. The stage 1 model assumes the WS variance to be homogeneous. The stage 2 model adds in the WS variance covariates. The stage 3 model adds in the random-scale effects. The linear or quadratic subject-level association between the random-location effects and the WS variance is also added at this stage if it has been specified. In the current model, we specify there to be no such association and so the only difference between the stage
2 and stage 3 models is the inclusion of the random-scale effects. We can therefore simply compare the stage 2 and stage 3 model results for WS variance depression group regression coefficient.

The stage 1, 2 and 3 estimation results are saved by runmixregls and can be listed using the ereturn list command. Full explanations are given in the runmixregls help file.

. ereturn list

scalars:

 Scalars:
  e(N) = 375
  e(N_g) = 66
  e(g_min) = 4
  e(g_avg) = 5.681818181818182
  e(g_max) = 6
  e(k) = 10
  e(k_1) = 7
  e(k_2) = 9
  e(k_3) = 10
  e(time) = 8.561999999999999
  e(ll) = -1123.351440429688
  e(ll_1) = -1140.599487304688
  e(ll_2) = -1134.499755859375
  e(ll_3) = -1123.351440429688
  e(iterations) = 14
  e(iterations_1) = 5
  e(iterations_2) = 3
  e(iterations_3) = 6
  e(chi2_1vs2) = 12.19946289062591
  e(chi2_1vs3) = 34.49609375
  e(chi2_2vs3) = 2.336161284000e-06
  e(p_2vs3) = 1.55659769135e-07
  e(p_1vs3) = .0022434701330112
  e(p_1vs2) = 22.29663085937409

 macros:
  e(cmd) : "runmixregls"
  e(title) : "Mixed-effects location scale model"
  e(depvar) : "hamdep"
  e(ivar) : "id"
  e(adapt) : "1"
  e(n_quad) : "15"
  e(iterate) : "200"
  e(tolerance) : ".0005"
  e(standardize) : "0"
  e(properties) : "b V"

 matrices:
functions:

e(sample)

The names assigned to the different stage specific model results are suffixed by _1, _2 and _3. For example, the parameter estimates from each stage are stored in the matrices e(b_1), e(b_2) and e(b_3), while their corresponding variance-covariance matrices are stored in the matrices e(V_1), e(V_2) and e(V_3).

We can examine the stage 2 parameter estimates by using the matrix list command.

```
. matrix list e(b_2)
e(b_2)[1,9]
   Mean: Mean: Mean: Mean: Between: Between:
     week endog endweek _cons endog _cons
  y1  -2.3985558  1.8533485  .01527996  22.55652  .48166201  2.2502859
   Within: Within: Within:
     week endog _cons
  y1   .17670505  .27196762  2.3461366
```

More usefully, we can construct a complete estimates table for the stage 2 results by first making copies of e(b_2) and e(V_2).

```
. matrix b_2 = e(b_2)
. matrix V_2 = e(V_2)
```

Then we store these copies using the ereturn post command.

```
. ereturn post b_2 V_2
```

Finally we display the complete estimates table using the ereturn display command.

```
. ereturn display
```

| Coef.  Std. Err.  z  P>|z|  [95% Conf. Interval] |
|-------|-----------------|--------|--------|------------------|
| Mean  |                 |        |        |                  |

The group difference in the WS variance now appears significant at the 10% level ($\hat{\tau}_2 = 0.27$, $p = 0.09$), whereas when we accounted for the random-scale effects it was not significant even at the 20% level. In Section 7 we shall demonstrate that the apparent increase in the precision of the WS variance function coefficients when ignoring the random-scale effects is entirely spurious.

### 6. Example 2: Positive mood study

The second example analyses presented by Hedeker and Nordgren (2013) relate to data drawn from a week-long intensive longitudinal study of 515 adolescent school students (Mermelstein, Hedeker, Flay, and Shiman 2002). Ecological momentary assessments (EMA) were collected by randomly prompting students to respond to questions about their mood, place, activity and companionship. The data are two-level with observations (level-1) nested within students (level-2). The response is a positive mood score and interest lies in explaining why subjects differ in their positive mood variability from prompt-to-prompt as well as why some subjects have higher average positive mood than others. The main covariates of interest are student gender and whether the student was alone when prompted.

#### 6.1. Data

First we load the data.

```
. use posmood, clear
```

Next we describe the data’s content.

```
. codebook, compact
  Variable Obs Unique Mean Min Max Label
  +-----------------------------------------------
  subject 17514 515 257.635 1 515 Subject
  posmood 17514 28 6.732652 1 10 Positive mood score
```
alone  17514  2  .4627155  0  1  Alone
genderf  17514  2  .5842183  0  1  Female

The output shows that the data includes four variables: the subject identifier, subject; the response variable, posmood (range = 1 to 10); the week number, week (coded sequentially from 0 to 5); whether the student was alone when prompted, alone (coded 0 for with others, 1 for alone); and student gender, genderf (coded 0 for males, 1 for females). In total the 515 subjects contributed 17514 observations. The mean positive mood is 6.73 out of 10 and subjects were alone with a proportion of 0.46 or 46% of the observations.

To get a better sense of the data, Figure 4 plots a trellis plot of positive mood scores against observation number for the first 20 subjects in the data. To plot this figure, we must first generate a unique observation identifier within each subject and then xtset the data.

. bysort subject: gen observation = _n
. label var observation "Observation"
. xtset subject observation
    panel variable: subject (unbalanced)
    time variable: observation, 1 to 58
    delta: 1 unit

The xtset output informs us that at least one subject contributes as many as 58 observations. Next we use the xtline command to draw the plot.

. xtline posmood if subject<=20, byopts(style(compact))

where we specify if subject<=20 to restrict the plot to the first 20 subjects and we specify the byopts(style(compact)) option to draw a more compact version of the plot than is plotted by default.

The plot shows some subjects contribute more observations than others. We also see that both mean positivity and mood variability varies substantially across subjects. For example, Subjects 17 and 19 show similar mean positivity, but differ substantially in their mood variability with Subject 19 exhibiting a more variable mood than Subject 17. In contrast, Subject 18 exhibits very high mean positivity, close to the maximum score of 10, and very low mood variability. This pattern is consistent with a ceiling effect in the positive mood scale.

6.2. Model 1

The first model Hedeker and Nordgren (2013, p. 22) specify for these data is written as

\[
\begin{align*}
\text{posmood}_{ij} &= \beta_0 + \beta_1 \text{alone}_{ij} + \beta_2 \text{genderf}_i + \sigma_{\text{ui}_i} \theta_{1i} + \epsilon_{ij}, \\
\log\left(\sigma^2_{\text{ui}_i}\right) &= \alpha_0 + \alpha_1 \text{alone}_{ij} + \alpha_2 \text{genderf}_i, \\
\log\left(\sigma^2_{\epsilon_{ij}}\right) &= \tau_0 + \tau_1 \text{alone}_{ij} + \tau_2 \text{genderf}_i + \tau_3 \theta_{1i} + \sigma_\omega \theta_{2i}, \\
\theta_{1i} &\sim \mathcal{N}(0, 1), \\
\theta_{2i} &\sim \mathcal{N}(0, 1), \\
\epsilon_{ij} &\sim \mathcal{N}\left(0, \sigma^2_{\epsilon_{ij}}\right),
\end{align*}
\]

(16) (17) (18) (19) (20) (21)
where in the mean function posmood is modeled in terms of alone and genderf, to allow for mean mood differences by gender and whether subjects are alone. The random-location effect, $\sigma_{\theta_1 i}$, allows for subject intercept heterogeneity above and beyond that explained by the covariates. The log of the BS variance, $\sigma^2_{\epsilon_{ij}}$, is modeled as a function of genderf to allow male and female subjects to be differentially variable in terms of their mean positive mood scores, having adjusted for the mean differences in gender and whether subjects are alone. It is also modeled as a function of alone and so the magnitude of each subject’s random-location effect is allowed to be modified on occasions when they are alone. The log of the WS variance, $\sigma^2_{\epsilon_{ij}}$, is modeled as a function of alone and genderf to allow the variability of a subject’s positive mood about their individual mean level to differ by gender and whether students are alone. The location effect is entered into the WS variance function to allow for a linear subject-level association between the log of the WS variance and the random-location effect. The random-scale effect, $\sigma_{\omega_{2i}}$, allows for any remaining unexplained variation in WS response heterogeneity across subjects.

```
. runmixregls posmood alone genderf, between(alone genderf) > within(alone genderf)
```

```
runmixregls - Run MIXREGLS from within Stata

Mixed-effects location scale model  Number of obs =  17514
Group variable: subject  Number of groups =  515
Obs per group: min =  3
               avg =  34.0
               max =  58
```
Run time (seconds) = 317.061
Integration points = 11
Log Likelihood = -34141.961

<table>
<thead>
<tr>
<th>posmood</th>
<th>Coef. Std. Err. z P&gt;</th>
<th>z</th>
<th>[95% Conf. Interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mean</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>-.3699634 .025034 -14.78 0.000 -.4190291 -.3208977</td>
<td></td>
<td></td>
</tr>
<tr>
<td>genderf</td>
<td>-.1500064 .1090527 -1.38 0.169 -.3637458 .0637331</td>
<td></td>
<td></td>
</tr>
<tr>
<td>_cons</td>
<td>6.990347 .0810537 86.24 0.000 6.831485 7.149209</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Between</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>.1053504 .0364417 2.89 0.004 .033926 .1767749</td>
<td></td>
<td></td>
</tr>
<tr>
<td>genderf</td>
<td>.0044638 .1305573 0.03 0.973 -.2514238 .2603513</td>
<td></td>
<td></td>
</tr>
<tr>
<td>_cons</td>
<td>.2984207 .0978013 3.05 0.002 .1067336 .4901078</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Within</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>.0807709 .0247427 3.26 0.001 .0322762 .1292657</td>
<td></td>
<td></td>
</tr>
<tr>
<td>genderf</td>
<td>.2159367 .0613066 3.52 0.000 .0957779 .3360954</td>
<td></td>
<td></td>
</tr>
<tr>
<td>_cons</td>
<td>.7632344 .0474795 16.08 0.000 .6701763 .8562924</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Association</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>linear</td>
<td>-.217607 .0310226 -7.01 0.000 -.2784102 -.1568039</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Scale</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>sigma</td>
<td>.597438 .022634 26.40 0.000 .5530761 .6418</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

The summary information at the top of the output states that the model is fitted to an estimation sample of 17514 observations belonging to 515 subjects. The number of observations per subject ranges from 3 to 58 with an average of 34 observations. The model took 317 seconds to converge.

The substantive interpretation of these results is given by Hedeker and Nordgren (2013, p. 23), and so we provide only a brief summary here. The mean function shows subjects experience significantly lower moods when alone ($\hat{\beta}_1 = -0.37$, $p < 0.05$), but there is no mean mood difference by gender ($\hat{\beta}_2 = -0.15$, $p = 0.17$). The BS variance function also shows females to be no more variable than males in their mean levels of positivity ($\hat{\alpha}_2 = 0.004$, $p = 0.97$). However, unexplained differences in mean moods between subjects become more pronounced when subjects are alone relative to when they are with others ($\hat{\alpha}_1 = 0.11$, $p < 0.05$). The WS variance function shows females have significantly more variable moods from one observation to the next than males ($\hat{\tau}_2 = 0.22$, $p < 0.05$). Subjects are also shown to have more variable observation-to-observation moods when they are alone relative to when they are with others ($\hat{\tau}_1 = 0.08$, $p < 0.05$). The linear random-location effects are negatively associated with the WS variance and so, having adjusted for the covariates, subjects with higher mean moods tend to have lower mood variability ($\hat{\tau}_l = -0.22$, $p < 0.05$). Finally, the random-scale standard deviation is significant suggesting that there is an unexplained component to subjects’ mood variability ($\hat{\sigma}_\omega = 0.60$, $p < 0.05$).
We store these estimation results, giving them the name *model1*, so that we can retrieve them at a later point.

```
.estimates store model1
```

To aid the interpretation of this model Hedeker and Nordgren (2013, p. 25) predict the BS variance and the WS variance separately for the four subject groups formed by crossing gender by alone status: (1) Male, not alone; (2) Male, alone; (3) Female, not alone; (4) Female, alone. Below we replicate these predictions. First we use the `generate` and `replace` commands with appropriate if statements to generate a new categorical variable named `group` which indicates these four subject groups.

```
.generate group = .
(17514 missing values generated)
.replace group = 1 if genderf==0 & alone==0
(3810 real changes made)
.replace group = 2 if genderf==0 & alone==1
(3472 real changes made)
.replace group = 3 if genderf==1 & alone==0
(5600 real changes made)
.replace group = 4 if genderf==1 & alone==1
(4632 real changes made)
```

Then we define a value label named `grouplabel` and attach this to this variable.

```
.label define grouplabel 1 "Male, not alone"
> 2 "Male, alone"
> 3 "Female, not alone"
> 4 "Female, alone"
.label values group grouplabel
```

Taking the exponential of both sides of the log-linear BS variance function (Equation 17) gives

\[
\sigma^2_{ij} = \exp(\alpha_0 + \alpha_1\text{alone}_{ij} + \alpha_2\text{gender}_i).
\]  

We use the `predictnl` command to store these predictions in a new variable called `BSvariance`. We make use of the function `xb(Between)` which replicates the calculation of the linear predictor, $u_{ij}\hat{\alpha}$, here, $\hat{\alpha}_0 + \hat{\alpha}_1\text{alone}_{ij} + \hat{\alpha}_2\text{gender}_i$. 


. predictnl BSvariance = exp(xb(Between))

Next we use the \texttt{tabstat} command to tabulate the predicted BS variance by subject group.

. tabstat BSvariance, by(group) nototal format(%4.3f)

Summary for variables: BSvariance
by categories of: group

<table>
<thead>
<tr>
<th>group</th>
<th>mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male, not alone</td>
<td>1.348</td>
</tr>
<tr>
<td>Male, alone</td>
<td>1.497</td>
</tr>
<tr>
<td>Female, not alone</td>
<td>1.354</td>
</tr>
<tr>
<td>Female, alone</td>
<td>1.504</td>
</tr>
</tbody>
</table>

These predicted values show the variability in mean moods across subjects are amplified when subjects are alone, relative to when they are with others.

Taking the exponential of both sides of the log-linear WS variance function (Equation 18) gives

\[
\sigma_{\epsilon ij}^2 = \exp \left( \tau_0 + \tau_{\text{alone}}ij + \tau_{\text{genderf}}i + \tau_l \theta_1i + \sigma_{\omega}^2 \theta_2i \right). \tag{23}
\]

We wish to obtain the population-averaged WS variance for each of our four subject groups. This is obtained by integrating out the random-location and random-scale effects. It can be shown that the resulting expression is given by

\[
\sigma_{\epsilon ij}^2 = \exp \left\{ \tau_0 + \tau_{\text{alone}}ij + \tau_{\text{genderf}}i + \frac{1}{2} \left( \tau_l^2 + \sigma_{\omega}^2 \right) \right\}. \tag{24}
\]

We will use the \texttt{predictnl} command to also store these predictions. However, in order to do so we first need to know how to specify the parameters $\tau_l$ and $\sigma_\omega$ in the above expression. We can find their internal Stata parameter names by reissuing the \texttt{runmixregls} command with only the \texttt{coeflegend} option. This will simply redisplay the current (active) estimation results, but will present the parameter names rather than displaying the statistics for these coefficients.

. runmixregls, coeflegend

\texttt{runmixregls} - Run MIXREGLS from within Stata

Mixed-effects location scale model
Group variable: subject
Number of obs = 17514
Number of groups = 515
Obs per group: min = 3
avg = 34.0
max = 58

\footnote{We integrate out the random-scale effects by making use of the fact that if a random variable is log normally distributed with mean $\mu$ and variance $\sigma^2$, then its expected value is given by $\exp (\mu + \frac{1}{2} \sigma^2)$.}
runmixregls: Run MIXREGLS from within Stata

Run time (seconds) = 317.852
Integration points = 11
Log Likelihood = -34141.961

<table>
<thead>
<tr>
<th>Coef.</th>
<th>Legend</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mean</td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>-.3699634 _b[Mean:alone]</td>
</tr>
<tr>
<td>genderf</td>
<td>-.1500064 _b[Mean:genderf]</td>
</tr>
<tr>
<td>_cons</td>
<td>6.990347 _b[Mean:_cons]</td>
</tr>
<tr>
<td>Between</td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>.1053504 _b[Between:alone]</td>
</tr>
<tr>
<td>genderf</td>
<td>.0044638 _b[Between:genderf]</td>
</tr>
<tr>
<td>_cons</td>
<td>.2984207 _b[Between:_cons]</td>
</tr>
<tr>
<td>Within</td>
<td></td>
</tr>
<tr>
<td>alone</td>
<td>.0807709 _b[Within:alone]</td>
</tr>
<tr>
<td>genderf</td>
<td>.2159367 _b[Within:genderf]</td>
</tr>
<tr>
<td>_cons</td>
<td>.7632344 _b[Within:_cons]</td>
</tr>
<tr>
<td>Association</td>
<td></td>
</tr>
<tr>
<td>linear</td>
<td>-.217607 _b[Association:linear]</td>
</tr>
<tr>
<td>Scale</td>
<td></td>
</tr>
<tr>
<td>sigma</td>
<td>.597438 _b[Scale:sigma]</td>
</tr>
</tbody>
</table>

We see that the internal Stata parameter names for \( \tau \) and \( \sigma_\omega \) are \_b[Association:linear] and \_b[Scale:sigma], respectively. We can now use `predictnl` to predict the population-averaged WS variances, placing them in a new variable called `WSvariance`. We make use of the function `xb(Within)` to replicate the calculation of the linear predictor, \( \mathbf{w}_{ij}^\top \hat{\mathbf{\tau}}, \) here, \( \hat{\tau}_0 + \hat{\tau}_{1 \text{alone}}_{ij} + \hat{\tau}_{2 \text{genderf}}. \)

```
. predictnl WSvariance = exp(xb(Within) + 0.5 * (_b[Association:linear]^2 + _b[Scale:sigma]^2))
```

Next we tabulate the predicted population-averaged WS variance by subject group.

```
. tabstat WSvariance, by(group) nototal format(%4.3f)
```

Summary for variables: WSvariance
   by categories of: group

<table>
<thead>
<tr>
<th>group</th>
<th>mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male, not alone</td>
<td>2.626</td>
</tr>
</tbody>
</table>
Male, alone | 2.847  
Female, not alone | 3.259  
Female, alone | 3.533  

These predicted values show that females have higher occasion-to-occasion mood variability than males. Similarly subjects have more variable moods when alone, relative to when they are with others.

Having predicted the BS and WS variances for each subject group, Hedeker and Nordgren (2013, p. 25) go on to calculate the intraclass correlation coefficient (ICC) for each subject group. The ICC is defined as the expected residual correlation between two observations from the same subject and is calculated as $r = \frac{\sigma^2_{v_{ij}}}{\left(\sigma^2_{v_{ij}} + \sigma^2_{\epsilon_{ij}}\right)}$.

We can once again use the `predictnl` command to calculate these predictions:

```
predictnl ICC = exp(xb(Between))/(exp(xb(Between)) + exp(xb(Within))  
               > + 0.5 * (_b[Association:linear]^2 + _b[Scale:sigma]^2))
```

We then tabulate the ICC by subject group:

```
. tabstat ICC, by(group) nototal format(%4.3f)
```

Summary for variables: ICC  
by categories of: group

<table>
<thead>
<tr>
<th>group</th>
<th>mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>Male, not alone</td>
<td>0.339</td>
</tr>
<tr>
<td>Male, alone</td>
<td>0.345</td>
</tr>
<tr>
<td>Female, not alone</td>
<td>0.294</td>
</tr>
<tr>
<td>Female, alone</td>
<td>0.299</td>
</tr>
</tbody>
</table>

These predicted values show that there is substantial clustering in the data for all four groups, but that observations are more correlated within males than females. In contrast, subjects’ observations appear no more correlated when they are alone, than when they are with others.

6.3. Model 2

The second model Hedeker and Nordgren (2013, p. 26) specify explores the interaction between gender and whether subjects are alone on their mean moods, BS heterogeneity in mean moods, and in their WS mood variability. Specifically, they include `algenf`, the interaction between `alone` and `genderf` in the mean function, the BS variance function, and the WS variance.
function. Their model can be written as

\[ \text{posmood}_{ij} = \beta_0 + \beta_1 \text{alone}_{ij} + \beta_2 \text{genderf}_i + \beta_3 \text{algenf}_{ij} + \sigma_{\epsilon_{ij}} \theta_{1i} + \epsilon_{ij}, \quad (25) \]

\[ \log \left( \sigma_{\epsilon_{ij}}^2 \right) = \alpha_0 + \alpha_1 \text{alone}_{ij} + \alpha_2 \text{genderf}_i + \alpha_3 \text{algenf}_{ij}, \quad (26) \]

\[ \log \left( \sigma_{\nu_{ij}}^2 \right) = \tau_0 + \tau_1 \text{alone}_{ij} + \tau_2 \text{genderf}_i + \tau_3 \text{algenf}_{ij} + \tau_4 \theta_{1i} + \sigma_\omega \theta_{2i}, \quad (27) \]

\[ \theta_{1i} \sim \mathcal{N}(0, 1), \quad (28) \]

\[ \theta_{2i} \sim \mathcal{N}(0, 1), \quad (29) \]

\[ \epsilon_{ij} \sim \mathcal{N}\left(0, \sigma_{\epsilon_{ij}}^2\right). \quad (30) \]

First we generate the interaction term.

\[ . \ generate \ algenf = \text{alone} \times \text{genderf} \]

Then we fit the model.

\[ . \ runmixregls \ posmood \ alone \ genderf \ algenf, \ between(\text{alone} \ genderf \ algenf) \]

\[ > \ within(\text{alone} \ genderf \ algenf) \]

runmixregls - Run MIXREGLS from within Stata

Mixed-effects location scale model Number of obs = 17514
Group variable: subject Number of groups = 515
Obs per group: min = 3
avg = 34.0
max = 58

Run time (seconds) = 249.67
Integration points = 11
Log Likelihood = -34137.281

| posmood | Coef. | Std. Err. | z  | P>|z| | [95% Conf. Interval] |
|---------|-------|-----------|----|------|----------------------|
| Mean    |       |           |    |      |                      |
| alone   | -0.2931987 | 0.037085 | -7.91 | 0.000 | -0.3658841 to -0.2205134 |
| genderf | -0.1228584 | 0.1095034 | -1.12 | 0.262 | -0.3374813 to 0.0917644 |
| algenf  | -0.1435512 | 0.0502354 | -2.86 | 0.004 | -0.2420109 to -0.0450916 |
| _cons   | 6.975568  | 0.0815586 | 85.53 | 0.000 | 6.815716 to 7.13542    |

<table>
<thead>
<tr>
<th>Between</th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>alone</td>
<td>0.0954204</td>
<td>0.050034</td>
<td>1.91</td>
<td>0.057</td>
<td>-0.0026444 to 0.1934852</td>
</tr>
<tr>
<td>genderf</td>
<td>0.005984</td>
<td>0.135521</td>
<td>0.04</td>
<td>0.665</td>
<td>-0.2593013 to 0.2652903</td>
</tr>
<tr>
<td>algenf</td>
<td>-0.0091599</td>
<td>0.0733212</td>
<td>-0.12</td>
<td>0.901</td>
<td>-0.1528669 to 0.134547</td>
</tr>
<tr>
<td>_cons</td>
<td>0.3057143</td>
<td>0.099226</td>
<td>3.08</td>
<td>0.002</td>
<td>0.112349 to 0.5001937</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Within</th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>alone</td>
<td>0.0995315</td>
<td>0.0384968</td>
<td>2.59</td>
<td>0.010</td>
<td>0.0240792 to 0.1749837</td>
</tr>
</tbody>
</table>
We store the estimation results, naming them `model2`.

```
. estimates store model2
```

We can replicate the likelihood ratio test comparing this model to the previous one (Hedeker and Nordgren 2013, p. 27) by using Stata’s `lrtest` command.

```
. lrtest model1 model2
```

Likelihood-ratio test LR chi2(3) = 9.36
(Assumption: model1 nested in model2) Prob > chi2 = 0.0249

We see that there is evidence of a significant interaction between gender and whether subjects are alone on positive mood ($\chi^2_3 = 9.36, p = 0.03$). However, examining the individual parameters shows that only the mean function interaction is individually significant ($\hat{\beta}_3 = -0.14, p < 0.05$). The nature of this interaction is that there is no gender difference in positivity when subjects are with others ($\hat{\beta}_2 = -0.12, p = 0.26$) but a gender difference becomes apparent when subjects are alone with females exhibiting significantly lower mood than males ($\hat{\beta}_2 + \hat{\beta}_3 = -0.27, p = 0.02$). We calculated the $p$ value for this result using the `lincom` command which computes point estimates, standard errors, $z$ statistics, $p$ values, and confidence intervals for linear combinations of model parameters. We refer to the estimated parameters $\hat{\beta}_2$ and $\hat{\beta}_3$ by their internal Stata parameter names.

```
. lincom _b[Mean:genderf] + _b[Mean:algenf]
```

( 1) [Mean]genderf + [Mean]algenf = 0

```
          posmood |  Coef.  Std. Err.     z  P>|z|   [95% Conf. Interval]
-------------+--------------------------------------------------
(1) | -.2664096    .1152927  -2.31    0.021   -.492379   -.0404401
```

### 6.4. Model 3

The third and final model Hedeker and Nordgren (2013, p. 28) specify allows for a quadratic relationship between the log of a subject’s WS variance and their random-location effect.
Such functionality will better capture the suspected ceiling effect in these data. Their model can be written as

\[
\text{posmood}_{ij} = \beta_0 + \beta_1 \text{alone}_{ij} + \beta_2 \text{genderf}_i + \beta_3 \text{algenf}_{ij} + \sigma_{\epsilon_{ij}} \theta_{1i} + \epsilon_{ij},
\]

\[
\log \left( \sigma^2_{\epsilon_{ij}} \right) = \alpha_0 + \alpha_1 \text{alone}_{ij} + \alpha_2 \text{genderf}_i + \alpha_3 \text{algenf}_{ij},
\]

\[
\log \left( \sigma^2_{\theta_{1i}} \right) = \tau_0 + \tau_1 \text{alone}_{ij} + \tau_2 \text{genderf}_i + \tau_3 \text{algenf}_{ij} + \sigma_\omega \theta_2i + \sigma_{\omega}\theta_{2i},
\]

\[
\theta_{1i} \sim \mathcal{N}(0, 1),
\]

\[
\theta_{2i} \sim \mathcal{N}(0, 1),
\]

\[
\epsilon_{ij} \sim \mathcal{N} \left( 0, \sigma^2_{\epsilon_{ij}} \right).
\]

To fit the model we simply add the `association(quadratic)` option to the previous call of `runmixregls`.

```
. runmixregls posmood alone genderf algenf, between(alone genderf algenf) > within(alone genderf algenf) association(quadratic)
```

```
runmixregls - Run MIXREGLS from within Stata

Mixed-effects location scale model                        Number of obs   =  17514
Group variable: subject                                   Number of groups =      515
Obs per group: min =          3                        avg =    34.0
                         max =         58
Run time (seconds) = 263.606
Integration points =          11
Log Likelihood = -34118.125

------------------------------------------------------------------------------
posmood |      Coef.    Std. Err.      z    P>|z|     [95% Conf. Interval]
---------+-----------------------------------------------------------------------------
   Mean   |                                                                
    alone |  -.2946179   .0374313   -7.87   0.000    -.3679818    -.221254
   genderf|  -.0542027   .1034123   -0.52   0.600    -.256887    .1484816
   algenf |  -.1371417   .0503696   -2.72   0.006   -.2358644   -.0384191
     _cons |   6.93622    .0784818   88.38   0.000     6.782399    7.090042
---------+-----------------------------------------------------------------------------
   Between |                                                                
    alone |   .0923244   .0485541    1.90   0.057    -.0028399    .1874886
   genderf|   .0354973   .1276911    0.28   0.781   -.2147725    .2857671
   algenf |  -.1371417   .0503696   -2.72   0.006   -.2358644   -.0384191
     _cons |   .2775077   .0958697    2.89   0.004    .0896065    .4654088
---------+-----------------------------------------------------------------------------
   Within  |                                                                
    alone |    .10084    .0385214    2.62   0.009     .0253394    .1763406
   genderf|   .2099701   .0650992    3.23   0.001     .0823781    .3375621
   algenf |  -.0323819   .0502295   -0.64   0.519   -.1308299    .066066
------------------------------------------------------------------------------
```
We store the estimation results, naming then `model3`.

```
. estimates store model3
```

We then perform a likelihood ratio test to compare this model to the previous model.

```
. lrtest model2 model3
```

Likelihood-ratio test       LR chi2(1) = 38.31
(Assumption: model2 nested in model3)       Prob > chi2 = 0.0000

We see that the current model, which includes the quadratic effect of the random-location effects on the WS variance, fits the data significantly better than the previous model which only allows for a linear effect ($\chi^2_1 = 38.31, p < 0.05$).

### 6.5. Additional analyses extending that presented in Hedeker and Nordgren (2013)

The `Association` section of the above estimates table shows that both the linear and quadratic effects of the random-location effects are negative and individually significant. We can aid our interpretation of these coefficients by plotting the WS variance function for each of the four subject groups as a function of these random-location effects where we integrate out the random-scale effects to simplify the plot. The resulting WS variance function is given by

$$
\sigma^2_{\epsilon_{ij}} = \exp \left( \tau_0 + \tau_1 \text{alone}_{ij} + \tau_2 \text{genderf}_i + \tau_3 \text{algenf}_{ij} + \tau_4 \theta_{1i} + \tau_5 \theta_{2i} + \frac{1}{2} \sigma^2_{\omega} \right). \quad (37)
$$

To plot this function separately for each of the four subject groups, we use the following `twoway` command with four separate `function` plots. The plottype `function` plots mathematical functions of the form $y = f(x)$, where $f(x)$ is some function of $x$. Here $x$ represents the standardized random-location effect, $\theta_{1i}$, and so we use the `range()` option to restrict its range to be from -3 to 3. We refer to all parameters by their internal Stata parameter names. We use the `ytitle()` and `xtitle()` options to specify the titles to appear next to the axes. We use the `legend()` option with the `order()` suboption to specify the text for the legend key.

```
. twoway ///
>   (function exp(_b[Within:_cons]
```
runmixregls: Run MIXREGLS from within Stata

Figure 5: Predicted WS variance functions plotted against random-location effect, separately for each subject group.

> + _b[Association:linear] * x + _b[Association:quadratic] * x^2
> + 0.5 * (_b[Scale:sigma]^2), range(-3 3))
> (function exp(_b[Within:_cons] + _b[Within:alone]
> + _b[Association:linear] * x + _b[Association:quadratic] * x^2
> + 0.5 * (_b[Scale:sigma]^2)), range(-3 3))
> (function exp(_b[Within:_cons] + _b[Within:genderf]
> + _b[Association:linear] * x + _b[Association:quadratic] * x^2
> + 0.5 * (_b[Scale:sigma]^2)), range(-3 3))
> (function exp(_b[Within:_cons] + _b[Within:alone]
> + _b[Within:genderf] + _b[Within:algenf]
> + _b[Association:linear] * x + _b[Association:quadratic] * x^2
> + 0.5 * (_b[Scale:sigma]^2)), range(-3 3)),
> ytitle(WS variance)
> xtitle(Random-location effect)
> legend(order(1 "Male, not alone"
> 2 "Male, alone"
> 3 "Female, not alone"
> 4 "Female, alone"))

Figure 5 shows females exhibit higher mood variability than males as do subjects of both genders when they are alone, relative to when they are with others. The negative linear and quadratic influences of the random-location effects give the variance functions their curved shapes and suggest subjects with more extreme location effects (particularly very high location
effects) exhibit less within-subject variation compared to those with less extreme location effects. This pattern reflects the floor and especially ceiling effects in the positive mood scale.

7. Simulation study

Most standard software can fit a limited version of the mixed-effects location scale model that allows heterogeneity in the WS variance across a limited number of sub groups. More specialized mixed-effects software such as HLM, MLwiN and SAS PROC MIXED can model the WS variance as a function of covariates. However, none of these packages allow users to additionally include random-scale effects in these functions. Ignoring unexplained subject differences in response dispersion will lead the resulting WS regression coefficients to be estimated with spurious precision, especially regression coefficients relating to subject-level predictors. This issue is analogous to the well known problem of ignoring clustering in standard mixed-effects models where omitting the random-location effects leads the mean function regression coefficients to be estimated with spurious precision. Despite this, we are not aware of this issue being discussed in the context of WS variance functions. Researchers ignoring unexplained subject variation in the residual error variance therefore run the risk of making type I errors of inference and of drawing misleading research conclusions regarding the factors which influence response dispersion. In Section 5 we showed using the Reisby data that ignoring the random-scale effects increased the apparent precision of the group difference in WS variance. In this section we present a simple parameter recovery simulation study to show that such increases in precision are entirely spurious.

We analyze 1000 replications of 250 subjects, with 10 observations per subject. On each replication, we generate subjects’ responses, \( y_{ij} \), according to the following “true” mixed-effects location scale model

\[
\begin{align*}
y_{ij} &= \beta_0 + \beta_1 x_{ij} + \beta_2 z_i + \sigma_v \theta_{1i} + \epsilon_{ij}, \\
\log(\sigma_v^2) &= \alpha_0, \\
\log\left(\frac{\sigma^2}{\sigma^2_{\epsilon_{ij}}}\right) &= \tau_0 + \tau_1 x_{ij} + \tau_2 z_i + \sigma_\omega \theta_{2i}, \\
\theta_{1i} &\sim \mathcal{N}(0, 1), \\
\theta_{2i} &\sim \mathcal{N}(0, 1), \\
\epsilon_{ij} &\sim \mathcal{N}\left(0, \sigma^2_{\epsilon_{ij}}\right),
\end{align*}
\]

(38) (39) (40) (41) (42) (43)

where the predictors, \( x_{ij} \) and \( z_i \), are standard normal variates defined at the observation and subject levels, respectively. For simplicity we assume a homogeneous BS variance and no subject-level association between the unobserved location and scale heterogeneity. The true values for the parameters are: \( \beta_0 = 0, \beta_1 = 0.5, \beta_2 = 0.2, \sigma_v = -2.3, \tau_0 = -0.61, \tau_1 = 0.1, \tau_2 = 0.1, \) and \( \sigma_\omega = 0.45 \). We then fit a mixed-effects location scale model to each simulated dataset which matches the true model specification presented above. We save the results of each stage 2 model (which ignores the random-scale effects) and each stage 3 model (which includes the random-scale effects). Lastly we compare the two sets of model results, averaging over the 1000 replications.

We start by clearing any existing data from memory and by specifying an initial value for the random-number seed should we wish to replicate our simulation study at a later date.
runmixregls: Run MIXREGLS from within Stata

. clear
. set seed 21561561

We then use the `forvalues` command to execute the commands enclosed within its braces 1000 times.

. forvalues r = 1/1000 {
    2.    display "Replication = `r'"
    3.    clear
    4.    set obs 250
    5.    generate subject = _n
    6.    generate z = rnormal(0, 1)
    7.    generate theta1 = rnormal(0, 1)
    8.    generate theta2 = rnormal(0, 1)
    9.    expand 10
   10.   bysort subject: generate obs = _n
   11.   generate x = rnormal(0, 1)
   12.   generate sigma_w = 0.45
   13.   generate sigma_v = sqrt(exp(-2.3))
   14.   generate sigma2_e = exp(-0.61 + 0.1 * x + 0.1 * z + sigma_w * theta2)
   15.   generate y = 0 + 0.5 * x + 0.2 * z + sigma_v * theta1 + e
   16.   xtset subject
   17.   runmixregls y x z, within(x z) association(none)
   18.   clear
   19.   set obs 8
   20.   generate byte parameter = _n
   21.   generate b_2 = e(b_2)'
   22.   generate V_2 = e(V_2)
   23.   generate V_3 = e(V_3)
   24.   svmat b_2
   25.   svmat se_2
   26.   svmat b_3
   27.   svmat se_3
   28.   rename (*1) (*)
   29.   format %4.3f b_2 se_2 b_3 se_3
   30.   list, abbreviate(9) separator(0)
   31.   save "rep_`r'.dta", replace
   32. }

(output omitted)

Line 2 of the loop displays the current replication number so that the user can see how far the simulation study has progressed. Line 3 clears any existing data from memory. Line 4 specifies that the new dataset will have 250 records, one for each subject. Line 5 generates a
new variable to index the subjects. Line 6 generates the single subject-level covariate. Lines 7 and 8 generate the standardized random-location and random-scale effects, respectively. Line 9 expands the data from one record per subject to ten records per subject, one for each observation. Line 10 generates a new variable to index these observations within each subject. Line 11 generates the single observation-level covariate. Lines 12 and 13 generate the BS standard deviations of the random-location and random-scale effects, respectively. Line 14 generates the WS variance. Line 15 generates the residual errors. Line 16 generates the response variable. Line 17 declares the data to be a panel. Line 18 fits the mixed-effects location scale model to the simulated data. Line 19 clears the data from memory. Line 20 changes the number of records in the dataset to eight, one for each parameter. Line 21 generates a new variable to index these parameters. Lines 22 through 25 create copies of the stage 2 and stage 3 coefficient vectors and their associated variance-covariance matrices. Lines 26 and 27 create column vectors of the stage 2 and stage 3 parameter standard errors as the square roots of the leading diagonals of their variance-covariance matrices. Lines 28 through 31 take the stage 2 and stage 3 column vectors of parameter estimates and standard errors and stores their values as four new variables. Line 32 removes the suffix 1 which is automatically added to the names of the four new variables. Line 33 sets the display precision of the four newly created variables to three decimal places. Line 34 lists the dataset of stage 2 and 3 model results. Line 35 saves these model results to disk, saving the replication number in the filename. Line 36 closes the loop.

Next we clear the data and generate a new variable called rep which will store the replication number.

```
. clear
. generate rep = .
```

We then append the 1000 datasets of model results stored on disk together.

```
. forvalues r = 1/1000 {
2. append using "rep_`r'"
3. replace rep = `r' if rep==.
4. }
```

(output omitted) Line 2 appends the dataset corresponding to the current replication number. Line 3 replaces missing values of rep with the current replication number in order to index the appended results. Line 4 closes the loop.

Next we generate a new variable called truevalue which holds the true value associated with each parameter

```
. generate truevalue = .
```

(8000 missing values generated)

```
. replace truevalue = 0.5 if parameter==1
```

(1000 real changes made)
. replace truevalue = 0.2 if parameter==2
(1000 real changes made)
. replace truevalue = 0.0 if parameter==3
(1000 real changes made)
. replace truevalue = -2.3 if parameter==4
(1000 real changes made)
. replace truevalue = 0.1 if parameter==5
(1000 real changes made)
. replace truevalue = 0.1 if parameter==6
(1000 real changes made)
. replace truevalue = -0.61 if parameter==7
(1000 real changes made)
. replace truevalue = 0.45 if parameter==8
(1000 real changes made)

Then we generate the percentage bias for each parameter making sure its display precision is to the nearest percentage point.

. generate b_2_bias = 100 * (b_2 - truevalue)/truevalue
(2000 missing values generated)
. generate b_3_bias = 100 * (b_3 - truevalue)/truevalue
(1000 missing values generated)
. format %3.0f b_2_bias b_3_bias

Then we generate a coverage indicator which equals 100 if the 95% confidence interval contains the true value, 0 otherwise. We use the inrange(z, a, b) function to return 1 if it is known that a ≤ z ≤ b; otherwise, the function returns 0.
. generate b_2_cov = 100 * (inrange(truevalue, b_2 - invnorm(0.975) * se_2, > b_2 + invnorm(0.975) * se_2))
. generate b_3_cov = 100 * (inrange(truevalue, b_3 - invnorm(0.975) * se_3, > b_3 + invnorm(0.975) * se_3))
. format %3.0f b_2_cov b_3_cov

Next we use the `collapse` command to average the results for each parameter over the 1000 replications. 

. collapse (mean) b_2 se_2 b_3 se_3 b_2_bias b_3_bias b_2_cov b_3_cov, > by(parameter truevalue)

We then define the value label `parlabel` and assign it to the variable `parameter`.

. label define parlabel 1 "beta_1" 2 "beta_2" 3 "beta_0" 4 "alpha_0" > 5 "tau_1" 6 "tau_2" 7 "tau_0" 8 "sigma_w"
. label values parameter parlabel

Finally we can display the results. First we list the averaged parameter estimates and their percentage biases. Note that the percentage bias for \( \beta_0 \) will be missing as its true value is zero.

. list parameter truevalue b_2 b_3 b_2_bias b_3_bias, noobs abbreviate(9) separator(0)

<table>
<thead>
<tr>
<th>parameter truevalue</th>
<th>b_2</th>
<th>b_3</th>
<th>b_2_bias</th>
<th>b_3_bias</th>
</tr>
</thead>
<tbody>
<tr>
<td>beta_1</td>
<td>.5</td>
<td>0.500</td>
<td>0.500</td>
<td>0</td>
</tr>
<tr>
<td>beta_2</td>
<td>.2</td>
<td>0.199</td>
<td>0.199</td>
<td>-0</td>
</tr>
<tr>
<td>beta_0</td>
<td>0</td>
<td>-0.001</td>
<td>-0.001</td>
<td>.</td>
</tr>
<tr>
<td>alpha_0</td>
<td>-2.3</td>
<td>-2.319</td>
<td>-2.319</td>
<td>1</td>
</tr>
<tr>
<td>tau_1</td>
<td>.1</td>
<td>0.101</td>
<td>0.101</td>
<td>1</td>
</tr>
<tr>
<td>tau_2</td>
<td>.1</td>
<td>0.101</td>
<td>0.101</td>
<td>1</td>
</tr>
<tr>
<td>tau_0</td>
<td>-.61</td>
<td>-0.508</td>
<td>-0.607</td>
<td>-17</td>
</tr>
<tr>
<td>sigma_w</td>
<td>.45</td>
<td>.</td>
<td>0.444</td>
<td>.</td>
</tr>
</tbody>
</table>

The stage 3 averaged parameter estimates (stored in `b_3`) lie very close to their true values (`truevalue`) and this is confirmed by their small percentage biases (`b_3_bias`). The stage 2 averaged parameter estimates (`b_2`) also lie very close to their true values, except for the intercept of the WS variance function, \( \tau_0 \), which is biased downwards by 17%. Recall that \( \tau_0 \) is expected to be lower in the stage 2 model as in that model \( \tau_0 \) corresponds to the population-averaged intercept while in the stage 3 model \( \tau_0 \) corresponds to the population-median intercept.

Second, we list the averaged standard errors.

. list parameter truevalue se_2 se_3, noobs abbreviate(9) separator(0)
runmixregls: Run MIXREGLS from within Stata

+---------------------------------------+
<table>
<thead>
<tr>
<th>parameter   truevalue    se_2     se_3</th>
</tr>
</thead>
<tbody>
<tr>
<td>beta_1     .5            0.016    0.015</td>
</tr>
<tr>
<td>beta_2     .2            0.025    0.025</td>
</tr>
<tr>
<td>beta_0     0             0.025    0.025</td>
</tr>
<tr>
<td>alpha_0    -2.3          0.146    0.144</td>
</tr>
<tr>
<td>tau_1      .1            0.030    0.032</td>
</tr>
<tr>
<td>tau_2      .1            0.030    0.042</td>
</tr>
<tr>
<td>tau_0      -.61          0.030    0.042</td>
</tr>
<tr>
<td>sigma_w    .45           .       0.044</td>
</tr>
</tbody>
</table>
+---------------------------------------+

The stage 2 averaged standard errors (stored in se_2) are very similar to the stage 3 averaged standard errors (se_3) for the mean function and the BS variance function parameters, but are smaller for the WS variance function parameters, especially for the coefficient of the subject-level covariate, $\tau_2$.

Finally, we examine the coverage rates.

. list parameter truevalue b_2_cov b_3_cov, noobs abbreviate(9) separator(0)

+---------------------------------------+
<table>
<thead>
<tr>
<th>parameter   truevalue    b_2_cov   b_3_cov</th>
</tr>
</thead>
<tbody>
<tr>
<td>beta_1     .5            96        96</td>
</tr>
<tr>
<td>beta_2     .2            94        94</td>
</tr>
<tr>
<td>beta_0     0             96        96</td>
</tr>
<tr>
<td>alpha_0    -2.3          96        97</td>
</tr>
<tr>
<td>tau_1      .1            92        95</td>
</tr>
<tr>
<td>tau_2      .1            82        95</td>
</tr>
<tr>
<td>tau_0      -.61          17        95</td>
</tr>
<tr>
<td>sigma_w    .45           100       96</td>
</tr>
</tbody>
</table>
+---------------------------------------+

The stage 3 coverage rates (stored in b_3_cov) show good coverage for all parameters; the 95% confidence intervals include the true value approximately 95% of the time. The stage 2 coverage rates (b_2_cov) show good coverage for the mean and BS variance functions parameters, but poor coverage for the WS variance function parameters. The coverage rate for the WS variance function intercept, $\tau_0$, is just 17% and is driven by the substantial downwards bias already described for this parameter. The coverage rate for the coefficient of the observation-level covariate, $\tau_1$, is 92%, while for the coefficient of the subject-level covariate, $\tau_2$, it is just 82%. Given that the corresponding averaged estimates show almost no bias for these two parameters, the low coverage rates shows that the standard errors in the stage 2 model are too small. In sum, ignoring the random-scale effect in the WS variance function leads the regression coefficients in this function to be estimated with spurious precision, especially the regression coefficients of subject-level covariates.
8. Conclusions

We have presented runmixregls, a command to run the MIXREGLS software for fitting mixed-effects location scale models seamlessly from within Stata. There are notable advantages of running MIXREGLS in this way. First, runmixregls provides a simple and intuitive command syntax to allow users to specify the full range of modeling and estimation options implemented in MIXREGLS. Second, prior to fitting a particular model, users can take advantage of Stata’s excellent statistics, graphics and data management commands to prepare and descriptively analyze their data. Third, after fitting the model, users can apply Stata’s comprehensive hypothesis testing, model comparison, prediction and plotting facilities to interpret their models and to communicate their results. Last, all analyses can be reproduced and documented for publication and review by typing all these commands into a file and running them all at once. We illustrated all this functionality by replicating and extending the two example analyses presented by Hedeker and Nordgren (2013). We also performed a simulation study, using runmixregls, to demonstrate a new and important research finding. Namely, that ignoring the random-scale effect in the WS variance function will lead to the regression coefficients in this function to be estimated with spurious precision, especially the regression coefficients of subject-level covariates.

The annotated Stata do-file which accompanies this article allows readers to replicate the presented analyses. A complete description of the command syntax and all modeling and estimation options is provided in the help file (Leckie 2013). Further information and resources can be found on the runmixregls web site (http://www.bristol.ac.uk/cmm/software/runmixregls/).

Acknowledgments

This research was funded under the LEMMA3 project, a node of the UK Economic and Social Research Council’s National Centre for Research Methods (grant number RES-576-25-0035).

References


Thompson J (2014). *Bayesian Analysis with Stata*. Stata Press, College Station, TX.


**Affiliation:**

George Leckie  
Centre for Multilevel Modelling  
Graduate School of Education  
University of Bristol  
2 Priory Road  
Bristol, BS8 1TX, United Kingdom  
E-mail: g.leckie@bristol.ac.uk  
URL: [http://www.bristol.ac.uk/cmm/team/leckie.html](http://www.bristol.ac.uk/cmm/team/leckie.html)